

ODS Magic: Using Lesser Known Features of the ODS statement

Michael Stout, J&J Medical Devices;

ABSTRACT

It is no allusion, the ODS output destination is a powerful tool that every SAS programmer should know how to use. This paper will provide details on lesser known aspects of ODS that are magical. With a slight of hand, this paper will show how to send desired SAS® output to multiple output destinations. Learn techniques to combine tables, listings and figures to a single output file. What you thought was impossible, may be easy using lesser known features of ODS.

INTRODUCTION

The Output Delivery System® (ODS) is truly magical. ODS provides a robust and easy to use method for rendering output to multiple output file types such as RTF, PDF, CSV, and Excel. ODS removes the complexity and greatly simplifies the process of delivering tables, listings and figures to customers.

It is not the intent of this paper to cover all aspects of ODS, but to focus on lesser known features of the ODS statement that provide significant value and flexibility in the way output is delivered to end users. This paper will primarily focus on the RTF output destination. The same techniques used for the RTF output destination can also be applied to other output destinations such as HTML.

An introduction and review of the ODS statement is presented to provide context and a general understanding the syntax. The remainder of the paper presents techniques for packaging or bundling output. After reading this paper you should be able to:

1. Send output to more than one ODS destinations
2. Send output to more than one ODS destination of the same type
3. Send multiple outputs to one ODS destination

Since this information presented is general in nature it is applicable to all skill levels, industries and job roles.

There is something here for everyone: from the novice to the seasoned SAS professional. It is always good to learn new “tricks”! I hope you will find the information presented in the paper useful and add it to your “bag of tricks”.

SYNTAX OF ODS STATEMENT

The ODS RTF destination is a fantastic choice for publishing tables, listings and figures generated using the SAS system. The general syntax is straight forward:

```
ODS <ODS destination>  
    <(identifier)>  
    <options>  
    <actions>;
```

The ODS statement has many options and actions. CLOSE, EXCLUDE, SELECT and SHOW are actions that may be applied to an ODS destination. Each action should appear on a separate ODS statement. When used, enclose identifiers in parentheses. A quick review of ODS options and actions presented in this paper include:

Options

ID=identifier	Unique instance of output destination. Choose a positive integer as the identifier.
Identifier	Shortcut. Same as ID=identifier
FILE=	Name of file to which output is written to
STYLE=	Name of styles template
SASDATE	Use the date and time when SAS session was started instead of default values in Word document
NOGTITLE	Titles associated with graphs appear like other output rendered to the RTF output destination. By default, titles are embedded with the graph.
NOGFOOTNOTE	Footnotes associated with graphs appear like other output rendered to the RTF output destination. By default, footnotes are embedded with the graph.

Actions

CLOSE	Close the RTF destination
SELECT=	Select output objects for the RTF destination
EXCLUDE=	Exclude output objects from the RTF destination
(PERSIST)	Works in conjunction with actions SELECT and EXCLUDE when running multiple procedures. This option is placed after each output object listed on SELECT and EXCLUDE actions.
SHOW	Displays the current object selection list or exclusion list in the SAS log

ODS TRACE is a convenient method to list the names of current objects in the SAS log. This tool is helpful in identifying object names that can be assigned to the SELECT and EXCLUDE options. Some commonly used object names include Report, Print, Sgplot, Summary and SQL_Results.

SEND OUTPUT TO MORE THAN ONE ODS DESTINATION

Abracadabra. The SAS output facility makes it easy to send output to multiple ODS destinations. The following example shows how to send PROC REPORT output to the RTF, HTML and EXCEL destinations:

```
%LET OUTDIR = \\itsusravmd10.jnj.com\group$\MDEV_BSDM\Users\Mstout2\_Test;

Title1 "Output Sent to more than one ODS destination";

ODS RTF FILE="&OUTDIR\_Report_EX1_RTF.rtf";
ODS HTML FILE="&OUTDIR\_Report_EX1_HTML.html";
ODS EXCEL FILE="&OUTDIR\_Report_EX1_EXCEL.xlsx";

PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway > 25));
```

```

RUN;

ODS RTF    CLOSE;
ODS HTML   CLOSE;
ODS EXCEL  CLOSE;
RUN;

```

SEND OUTPUT TO MORE THAN ONE ODS DESTINATION OF THE SAME TYPE

Sim Sala Bim. You may have a need to send output to multiple files with the same output destination. This may be handy if you want to send Report objects and Sgplot objects to separate RTF files. It's a snap using the ODS output destination. This is done using the ODS option called Identifier. The identifier is a unique integer that links objects to output destinations. You can write to a specific output destination by referencing the identifier:

```

%LET OUTDIR = \\itsusravmd10.jnj.com\group$\MDEV_BSDM\Users\Mstout2\_Test;

Title1 "Output Sent to more than one ODS destination of the same type";

ODS RTF (1) FILE="&OUTDIR\_Report_EX2_01.rtf";
ODS RTF (2) FILE="&OUTDIR\_Report_EX2_02.rtf";

/* Send REPORT object to RTF destination 1 */
ODS RTF (1) SELECT Report (PERSIST);

/* Send SGPLOT object to RTF destination 2 */
ODS RTF (2) SELECT Sgplot (PERSIST);

Title2 'Vehicles with Highway mileage up to 25 MPG';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway LE 25));
RUN;

PROC SGPLOT data=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway LE 25));
VBAR MPG_Highway;
RUN;

Title2 'Vehicles with Highway mileage over 25 MPG';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway > 25));
RUN;

PROC SGPLOT data=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway > 25));
VBAR MPG_Highway;
RUN;

ODS RTF (1) CLOSE;
ODS RTF (2) CLOSE;

RUN;

```

SEND MULTIPLE OUTPUTS TO ONE ODS DESTINATION

Hocus Pocus. Multiple outputs can be sent to the same output destination. This is a great way to package related outputs into a single file. The following code shows how to send output from three separate REPORT procedures to the same ODS destination:

```
%LET OUTDIR = \\itsusravmd10.jnj.com\group$\MDEV_BSDM\Users\Mstout2\_Test;

Title1 "Output Sent to more than one ODS destination of the same type";

ODS RTF FILE="&OUTDIR\_Report_EX3_Many2One.rtf";

TITLE2 'Ford vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway > 25));
RUN;

TITLE2 'Nissan vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Nissan' and MPG_Highway > 25));
RUN;

TITLE2 'Dodge vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Dodge' and MPG_Highway > 25));
RUN;

ODS RTF CLOSE;
RUN;
```

COMBINE CONCEPTS TO CREATE SOMETHING MAGICAL

Shazam. By combining the concepts presented in this paper, you will be able to create some magic in your organization. I have used these same concepts to create individual and comprehensive outputs with hundreds of pages. This example only keeps output objects created by the Report and SGPLOT procedures. The SELECT statement is a great way to limit the type and quantity of output sent to an ODS destination. Many SAS procedures create many output objects, but you may only care about some of them. The SELECT statement allows you to control which output objects are written to ODS destinations.

The SASDATE, STYLE, NOGTITLE and NOGFOOTNOTE options can be used to adjust the appearance of the RTF output.

Even if you like to create separate outputs for every Table, Listing and Figure, it is helpful to combine the output for internal review. This provides an opportunity for reviewers to add comments that facilitate changes and/or corrections for the entire deliverable. This improves traceability and helps SAS programming do their job more efficiently and accurately and most importantly improves communication. Communication is extremely important when working in a matrix organization or partnering with external vendors. The following code create four RTF files; 3 individual RTF files and one with all tables in a single RTF file:

```
%LET OUTDIR = \\itsusravmd10.jnj.com\group$\MDEV_BSDM\Users\Mstout2\_Test;

Title1 "Send output to individual AND combined files";
```

```

ODS RTF (99) FILE="&OUTDIR\_Report_EX4_99.rtf"
          SASDATE
          STYLE=styles.Journal
          NOGTITLE
          NOGFOOTNOTE
          ;

/* Only send REPORT and SGPOLT output to ODS destination 99 */
ODS RTF (99) SELECT Report (PERSIST) Sgplot (PERSIST);

ODS RTF FILE="&OUTDIR\_Report_EX4_01.rtf";
TITLE2 'Ford vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Ford' and MPG_Highway >
25));
RUN;
ODS RTF CLOSE;
RUN;

ODS RTF FILE="&OUTDIR\_Report_EX4_02.rtf";
TITLE2 'Nissan vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Nissan' and MPG_Highway >
25));
RUN;
ODS RTF CLOSE;
RUN;

ODS RTF FILE="&OUTDIR\_Report_EX4_03.rtf";
TITLE2 'Dodge vehicles with High Mileage';
PROC REPORT DATA=sashelp.CARS (WHERE=(Make='Dodge' and MPG_Highway >
25));
RUN;
ODS RTF CLOSE;
RUN;

ODS RTF (99) CLOSE;
RUN;

```

CONCLUSION

I hope you agree with me that these “tricks of the trade” are good tools to have in your “bag of tricks”. The Output Deliver System provide a great mechanism for publishing output in a variety of ways with relative ease. Use these tricks to improve communication, quality and reduce development time. Begin to use some of the lesser known features of the ODS to simply every-day tasks and do things you didn’t think were possible. ODS is truly magical and helps me every day to get the job done. Let the lesser known features of ODS help you too!

ACKNOWLEDGMENTS

I would like to thank my manager, Shuvo Rafiq, for his encouragement and support. This paper would not have been possible without his support.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Michael Stout
Mstout2@its.jnj.com